



# Electromechanical Switch Replacement


## Connecting Sensor Buttons to PIC12CXXX MCUs

Author: Vladimir Velchev  
AVEX  
Sofia, Bulgaria

### APPLICATION OPERATION

The idea is to replace the electromechanical switches with capacitive sensor buttons (PCB round or square pads). PIC12CXXX are very suitable for this purpose and with few components we may design a low cost fully electronic switch or regulator. The method uses a simple RC delay circuit (Figure 1), when the time constant changes when the sensor is touched.


**FIGURE 1: RC DELAY CIRCUIT**




To read the state of the sensor, the microcontroller must perform only two steps (Figures 2 and 3):

- Step 1: Changing the state of output from "0" to "1" (write operation -  $T_{wr}$ ).
- Step 2: Reading the state of input (read operation -  $T_{rd}$ ).

**FIGURE 2: UNTOUCHED SENSOR**



**FIGURE 3: UNTOUCHED SENSOR**



If the result of read operation is "0", it means that the sensor is touched. Human capacitance has been connected serially to capacitor C and the time constant of the circuit has become bigger.

Because of the small capacitance of human's fingers, the interval time between  $T_{wr}$  and  $T_{rd}$  should be less than 1-2  $\mu s$ , so we recommend the use of 4 MHz internal clock of PIC12CXXX.


The figure under Graphical Hardware Representation contains an example with two sensor buttons. The trick here is that for reading the sensor SB0: GP0 is set as input, GP1 is set as output and for reading the SB1: GP0 is set as output, GP1 is set as input. The outputs GP4,5 are connected to LEDs and show the state of sensor buttons. The other pins are not discussed here and they can be used for example as SYNCRO input and TRIAC DRIVER output.

The value of resistor R1 must be adjusted and it determines the sensibility of the buttons. The values of capacitors C1 and C2 are not so important. We recommend the use of the same type inputs of microcontroller, with equal input impedances.


Microchip Technology Incorporated, has been granted a nonexclusive, worldwide license to reproduce, publish and distribute all submitted materials, in either original or edited form. The author has affirmed that this work is an original, unpublished work and that he/she owns all rights to such work. All property rights, such as patents, copyrights and trademarks remain with author.

# Electromechanical Switch Replacement

## GRAPHICAL HARDWARE REPRESENTATION



## FLOWCHART



## MICROCHIP TOOLS USED

Assembler/Compiler Version:

MPLAB 3.22, MPASM 1.5

# Electromechanical Switch Replacement

## APPENDIX A: SOURCE CODE

```
;Software listing: sensor.asm
;*****
; Connecting sensor buttons to PIC12C508
; Osc.: F=4MHz (internal)
; Written by Vladimir Velchev 06.1997. (C) AVEX
; Version 1.00
;*****

; GP0 - sensor button SB0 (input/output)
; GP1 - sensor button SB1 (input/output)
; GP2 - not used (output)
; GP3 - not used (input)
; GP4 - LED1 for sensor button SB0 (output: 0=LED ON, 1=LED OFF)
; GP5 - LED2 for sensor button SB1 (output: 0=LED ON, 1=LED OFF)

LIST      P=12C508

#include <p12C508.inc>

;*** Equates
SB0          equ          0           ;sensor button SB0
SB1          equ          1           ;sensor button SB1
SB0_MASK     equ          B'00000001' ;bit mask for SB0
SB1_MASK     equ          B'00000010' ;bit mask for SB1
OUT0         equ          0           ;driving output for SB1
OUT1         equ          1           ;driving output for SB0
LED1         equ          4           ;LED for button SB0
LED2         equ          5           ;LED for button SB1
IOSET        equ          B'00001000' ;initial I/O port settings
                                   ;GP3-input, others- outputs

                                org          0           ;RESET vector

;*** Code Starting Point
MAIN:
; Initial setup
                                movlw       IOSET        ;init GPIO
                                tris        GPIO
                                clrf        GPIO          ;reset all outputs (=0)

Main_Loop:
                                clrwdt        ;clear watchdog timer
; Space for user code
;
; ...
; ...
; ...
                                call        READ_SENSORS ;call subroutine for sensors
                                btfss       STATUS,C      ;skip if sensor SB0 touched (C=1)
                                goto        SB0_untouched ;else - go to turn off LED1
                                bcf         GPIO,LED1      ;LED1=ON
                                goto        Check_SB1      ;go to checking next button
SB0_untouched:bsf        GPIO,LED1      ;LED1=OFF
Check_SB1:   btfss       STATUS,DC     ;skip if sensor SB1 touched (DC=1)
                                goto        SB1_untouched ;else - go to turn off LED2
                                bcf         GPIO,LED2      ;LED2=ON
                                goto        Main_Loop       ;go to beginning
SB1_untouched:bsf        GPIO,LED2      ;LED2=OFF
                                goto        Main_Loop       ;go to beginning

;*** Subroutine - READ_SENSORS
; Input :
; Output: Flags: C=1 if sensor SB0 touched, DC=1 if sensor SB1 touched
READ_SENSORS:
                                bcf         STATUS,C        ;clear carry flag
                                bcf         STATUS,DC       ;clear digit carry flag
```

# Electromechanical Switch Replacement

---

```
;Read sensor button SB0
    movlw        IOSET|SB0_MASK ;set SB0 as input (GP0)
    tris         GPIO
    bsf          GPIO,OUT1      ;send strobe to output ____
    btfss        GPIO,SB0       ;skip if sensor SB0 untouched (GP0=1)
    bsf          STATUS,C       ;else - set C flag
;Discharging the capacitance of the input
    movlw        IOSET&(~SB0_MASK) ;set SB0 as output (GP0)
    tris         GPIO
    bcf          GPIO,OUT0      ;reset output 0 (discharging)
    bcf          GPIO,OUT1      ;reset output 1
;Read sensor button SB1
    movlw        IOSET|SB1_MASK ;set SB1 as input (GP1)
    tris         GPIO
    bsf          GPIO,OUT0      ;send strobe to output ____
    btfss        GPIO,SB1       ;skip if sensor SB1 untouched (GP1=1)
    bsf          STATUS,DC      ;else - set DC flag
;Discharging the capacitance of the input
    movlw        IOSET&(~SB1_MASK) ;set SB1 as output (GP1)
    tris         GPIO
    bcf          GPIO,OUT1      ;reset output 1 (discharging)
    bcf          GPIO,OUT0      ;reset output 0
    return
end                                ;end of program
```

# Electromechanical Switch Replacement

---

NOTES:

# Electromechanical Switch Replacement

---

NOTES:

# Electromechanical Switch Replacement

---

NOTES:



**MICROCHIP**

---

## WORLDWIDE SALES & SERVICE

---

### AMERICAS

#### Corporate Office

Microchip Technology Inc.  
2355 West Chandler Blvd.  
Chandler, AZ 85224-6199  
Tel: 602-786-7200 Fax: 602-786-7277  
Technical Support: 602 786-7627  
Web: <http://www.microchip.com>

#### Atlanta

Microchip Technology Inc.  
500 Sugar Mill Road, Suite 200B  
Atlanta, GA 30350  
Tel: 770-640-0034 Fax: 770-640-0307

#### Boston

Microchip Technology Inc.  
5 Mount Royal Avenue  
Marlborough, MA 01752  
Tel: 508-480-9990 Fax: 508-480-8575

#### Chicago

Microchip Technology Inc.  
333 Pierce Road, Suite 180  
Itasca, IL 60143  
Tel: 630-285-0071 Fax: 630-285-0075

#### Dallas

Microchip Technology Inc.  
14651 Dallas Parkway, Suite 816  
Dallas, TX 75240-8809  
Tel: 972-991-7177 Fax: 972-991-8588

#### Dayton

Microchip Technology Inc.  
Two Prestige Place, Suite 150  
Miamisburg, OH 45342  
Tel: 937-291-1654 Fax: 937-291-9175

#### Los Angeles

Microchip Technology Inc.  
18201 Von Karman, Suite 1090  
Irvine, CA 92612  
Tel: 714-263-1888 Fax: 714-263-1338

#### New York

Microchip Technology Inc.  
150 Motor Parkway, Suite 416  
Hauppauge, NY 11788  
Tel: 516-273-5305 Fax: 516-273-5335

#### San Jose

Microchip Technology Inc.  
2107 North First Street, Suite 590  
San Jose, CA 95131  
Tel: 408-436-7950 Fax: 408-436-7955

#### Toronto

Microchip Technology Inc.  
5925 Airport Road, Suite 200  
Mississauga, Ontario L4V 1W1, Canada  
Tel: 905-405-6279 Fax: 905-405-6253

### ASIA/PACIFIC

#### Hong Kong

Microchip Asia Pacific  
RM 3801B, Tower Two  
Metroplaza  
223 Hing Fong Road  
Kwai Fong, N.T., Hong Kong  
Tel: 852-2-401-1200 Fax: 852-2-401-3431

#### India

Microchip Technology Inc.  
India Liaison Office  
No. 6, Legacy, Convent Road  
Bangalore 560 025, India  
Tel: 91-80-229-4036 Fax: 91-80-559-9840

#### Korea

Microchip Technology Korea  
168-1, Youngbo Bldg. 3 Floor  
Samsung-Dong, Kangnam-Ku  
Seoul, Korea  
Tel: 82-2-554-7200 Fax: 82-2-558-5934

#### Shanghai

Microchip Technology  
RM 406 Shanghai Golden Bridge Bldg.  
2077 Yan'an Road West, Hong Qiao District  
Shanghai, PRC 200335  
Tel: 86-21-6275-5700  
Fax: 86 21-6275-5060

#### Singapore

Microchip Technology Taiwan  
Singapore Branch  
200 Middle Road  
#07-02 Prime Centre  
Singapore 188980  
Tel: 65-334-8870 Fax: 65-334-8850

#### Taiwan, R.O.C

Microchip Technology Taiwan  
10F-1C 207  
Tung Hua North Road  
Taipei, Taiwan, ROC  
Tel: 886 2-717-7175 Fax: 886-2-545-0139

### EUROPE

#### United Kingdom

Arizona Microchip Technology Ltd.  
Unit 6, The Courtyard  
Meadow Bank, Furlong Road  
Bourne End, Buckinghamshire SL8 5AJ  
Tel: 44-1628-851077 Fax: 44-1628-850259

#### France

Arizona Microchip Technology SARL  
Zone Industrielle de la Bonde  
2 Rue du Buisson aux Fraises  
91300 Massy, France  
Tel: 33-1-69-53-63-20 Fax: 33-1-69-30-90-79

#### Germany

Arizona Microchip Technology GmbH  
Gustav-Heinemann-Ring 125  
D-81739 München, Germany  
Tel: 49-89-627-144 0 Fax: 49-89-627-144-44

#### Italy

Arizona Microchip Technology SRL  
Centro Direzionale Colleoni  
Palazzo Taurus 1 V. Le Colleoni 1  
20041 Agrate Brianza  
Milan, Italy  
Tel: 39-39-6899939 Fax: 39-39-6899883

### JAPAN

Microchip Technology Intl. Inc.  
Benex S-1 6F  
3-18-20, Shinyokohama  
Kohoku-Ku, Yokohama-shi  
Kanagawa 222 Japan  
Tel: 81-45-471- 6166 Fax: 81-45-471-6122

8/29/97

All rights reserved. © 1997, Microchip Technology Incorporated, USA. 9/97  Printed on recycled paper.

Information contained in this publication regarding device applications and the like is intended for suggestion only and may be superseded by updates. No representation or warranty is given and no liability is assumed by Microchip Technology Incorporated with respect to the accuracy or use of such information, or infringement of patents or other intellectual property rights arising from such use or otherwise. Use of Microchip's products as critical components in life support systems is not authorized except with express written approval by Microchip. No licenses are conveyed, implicitly or otherwise, under any intellectual property rights. The Microchip logo and name are registered trademarks of Microchip Technology Inc. in the U.S.A. and other countries. All rights reserved. All other trademarks mentioned herein are the property of their respective companies.